

excel yourself

Autoren: Ralf Sowa, Christian Hapke

Beachten Sie unsere [Hinweise](#) und [Nutzungsbedingungen](#). Vorgestellte Musterlösungen basieren auf MS-Excel® 2003; sie gelten ausschließlich für aufgezeigte Beispieldaten. Bitte melden Sie uns etwaige Fehler in unseren Informationen – Ihr Feedback ist willkommen: urs.toolbox@urs-beratung.de

Eine Übersicht zu unseren Excel-Informationen finden Sie hier: www.urs-beratung.de/toolbox.htm

Excel - Thema: BEREICH.VERSCHIEBEN

BEREICH.VERSCHIEBEN

BEREICH.VERSCHIEBEN

Mit der Funktion BEREICH.VERSCHIEBEN wird ein Bereich von Zellen definiert.

Definitionen

Die Funktion erwartet mindestens drei Argumente (in der folgenden Grafik fett geschrieben), zusätzlich ist die Angabe von zwei optionalen Parametern möglich:



(1) Bezug – die Zelle (oder der Zellenbereich), von dem aus betrachtet eine Verschiebung erfolgt

Im Beispiel ist Zelle A1 unser *Ausgangspunkt*.

(2) Zeilen – ist die Anzahl der Zeilen, um die eine Verschiebung des *Bezuges* erfolgt (positive Zahl = Verschiebung nach unten, negative Anzahl = Verschiebung nach oben)

Im Beispiel verschieben wir den Bezug um 4 Zeilen. Ausgehend von A1 befinden wir uns 4 Zeilen weiter unten in Zelle A5.

(3) Spalten – die Anzahl der Spalten, um die eine Verschiebung des *Bezuges* erfolgt (positive Zahl = Verschiebung nach rechts, negative Anzahl = Verschiebung nach links)

Im Beispiel verschieben wir den Bezug um 2 Spalten. Ausgehend von A5 befinden wir uns 2 Spalten weiter rechts in Zelle C5.

(4) Höhe – die Anzahl der betrachteten Zeilen des verschobenen *Bezuges*

Im Beispiel definieren wir die Höhe als 1 Zeile - damit bleibt der Bezug auf Zelle C5.

Weil schon der *Bezug* (A1) aus nur einer Zeile (nämlich Zeile 1) besteht - die Höhe des Bezuges bleibt also unverändert eine Zeile hoch - und die Angabe des Arguments *Höhe* optional ist, hätten wir hier (bei *Höhe*) gar nichts eintragen *brauchen*. Denn unterbleibt die Angabe für *Höhe* (gleiches gilt für *Breite*), umfasst der Bezug ebenso viele Zeilen, wie die Ausgangsbasis *Bezug*.

(5) Breite – die Anzahl der betrachteten Spalten des verschobenen Bezuges

Im Beispiel definieren wir die Breite als 3 Spalten - ausgehend von Zelle C5 umfasst unser Bereich also 3 Spalten, mithin die Zellen **C5:E5**

Wie bei vielen Funktionen in Excel, aber besonders bei dieser, ist es ratsam die Funktionsweise Schritt für Schritt in der Reihenfolge der gelieferten Funktionsargumente zu verstehen.

Wir ahnen, dass manchem unserer Leser obige Erläuterungen nicht genügen, die *ungewöhnliche* Funktion BEREICH.VERSCHIEBEN zu verstehen – für all diese unser ...

zweiter Erklärungsversuch

- Ausgangsbasis der Verschiebung und Dimensionierung

Das Funktionsargument **Bezug** stellt die Ausgangszelle(n) dar. Von hier aus (*Bezug*) startet das Verschieben (mit den Argumenten *Zeilen* und *Spalten*). Danach wird die Dimension (Anzahl der Zeilen und Spalten) des Bereiches (mit den Argumenten *Höhe* und *Breite*) bestimmt.

Wenn *Bezug* einen mehrere Zellen umfassenden Bereich beschreibt (z. B. D3:F5) ist die **erste Zelle des Bezuges** („oben links“) für das Verschieben maßgeblich – für den beispielhaft gewählten Bezug D3:F5 wäre also Zelle D3 maßgeblich.

- Verschieben des Bereiches

Das Verschieben erfolgt (ausschließlich) mit den Funktionsargumenten **Zeilen** und **Spalten**:

Zeilen verschiebt vertikal (nach unten/oben): positive Zahlen verschieben in höhere Zeilennummern, negative Zahlen in niedrigere Zeilennummern.

Spalten verschiebt horizontal (rechts/links) – positive Zahlen verschieben nach rechts, negative Zahlen nach links.

Betrachten Sie das Tabellenblatt als Koordinatensystem in dem Sie irgendwo ein Rechteck (das ist Ihr "Bezug") einzeichnen. Dieses Rechteck können Sie nun nach links/rechts bzw. oben/unten verschieben.

- Dimensionierung des Bereiches

Die Funktionsargumente **Höhe** und **Breite** dimensionieren den Bereich:

Höhe definiert die Anzahl der Zeilen.

Breite definiert die Anzahl der Spalten.

Die Funktionsargumente *Höhe* und *Breite* sind optional.

Erfolgt keine Angabe, bleibt die Dimension des Bereiches so groß wie unter *Bezug* definiert.

Werden *Höhe* und/oder *Breite* definiert, gilt die Angabe der Bereichsdimension für die erste Zelle „oben links“ des (mit den Funktionsargumenten *Zeilen* und *Spalten*) verschobenen (!) *Bezuges*.

Mit den Funktionsargumenten *Höhe* und *Breite* wird der Bereich nicht verkleinert oder vergrößert, sondern überhaupt **dimensioniert**. (Deshalb können diese Funktionsargumente auch nur positive Zahlen annehmen oder leer bleiben.**) Ausgangspunkt ist die erste Zelle des verschobenen Bereiches, die Dimensionierung bestimmt sich nach Angaben für die Argumente *Höhe* und *Breite* – und falls *Höhe* und/oder *Breite* fehlen, gelten die Dimensionen des *Bezuges*.

** Unsere Behauptung, die Funktionsargumente *Höhe* und *Breite* könnten nur positive Zahlen annehmen, ist nicht ganz korrekt: Zwar decken sich unsere Worte mit den Angaben in der Excel-Hilfe, tatsächlich aber arbeitet die Funktion auch mit negativen Zahlen für die Argumente *Höhe* und *Breite* – sie bewirken ein Dimensionieren des Bereichs nach oben bzw. nach links. Darauf gehen wir hier nicht näher ein (Interessierte finden in *Übung 5* dennoch ein Beispiel). Der Wert „Null“ ist für die Argumente *Höhe* und *Breite* übrigens auch *tatsächlich unzulässig*: er führt stets zur Fehlermeldung #BEZUG!. Das ist auch eingängig, denn ein Bereich kann nicht keine Zeilen oder keine Spalten beinhalten.

Übungen

Für unsere (*Trocken-*) *Übungen* benötigen Sie kein Excel. Bleiben Sie bitte einfach nur mit Ihrem Kopf bei uns. Im nächsten Kapitel *Erläuterungen* haben wir einige Beispiele vorbereitet, die Sie in Excel nachbilden können.

Übung 1 – positiver Zeilen-/Spaltenverschub ohne Änderung der Dimension

Funktionsargumente	
BEREICH.VERSCHIEBEN	
Bezug	D3:F5
Zeilen	1
Spalten	2
Höhe	
Breite	

Maßgeblich ist zunächst die erste Zelle von *Bezug*: Zelle D3.

Sie wird um 1 Zeile (nach D4) und um 2 Spalten (nach F4) verschoben. Der verschobene (Ziel-) Bereich beginnt also in Zelle F4.

Weil die Argumente *Höhe* und *Breite* nicht angegeben sind, bleibt die Dimension des Bereiches ebenso groß, wie unter *Bezug* angegeben – dort waren es 3 Spalten (D, E, F) und 3 Zeilen (3, 4, 5).

Der bei F4 beginnende Zielbereich ist also ebenfalls 3 Spalten (F, G, H) und 3 Zeilen (4, 5, 6) groß – Ergebnis: **F4:H6**

Übung 2 – negativer Zeilen-/Spaltenverschub ohne Änderung der Dimension

Funktionsargumente	
BEREICH.VERSCHIEBEN	
Bezug	D3:F5
Zeilen	-1
Spalten	-2
Höhe	
Breite	

Maßgeblich ist wieder die erste Zelle von *Bezug*: Zelle D3.

Sie wird um die **minus 1** Zeile (nach D2) und um **minus 2** Spalten (B2) verschoben. Der verschobene (Ziel-) Bereich beginnt also in Zelle B2.

Die Dimension des Bereiches bleibt wieder gleich groß. Der bei B2 beginnende Zielbereich ist wieder 3 Spalten (B, C, D) und 3 Zeilen (2, 3, 4) groß – Ergebnis: **B2:D4**

Übung 3 – Zeilen-/Spaltenverschub mit Definition der Dimension

Funktionsargumente	
BEREICH.VERSCHIEBEN	
Bezug	D3
Zeilen	1
Spalten	2
Höhe	3
Breite	3

Ausgangsbasis ist Zelle D3 aus *Bezug*.

Sie wird um 1 Zeile (nach D4) und um 2 Spalten (nach F4) verschoben. Der verschobene (Ziel-) Bereich umfasst bisher einzig die Zelle F4.


Das Argumente *Höhe* ist mit 3 angegeben: der Zielbereich umfasst also 3 Zeilen – ab Zelle F4 ist dies bisher der Bereich F4:F6.

Das Argument *Breite* ist ebenfalls mit 3 definiert – der Zielbereich umfasst also 3 Spalten – der Bereich F4:F6 wird also auf insgesamt 3 Spalten erweitert.

Der Zielbereich umfasst die Zellen: **F4:H6**

Vergleichen Sie hierzu das identische Ergebnis aus Übung 1!

Übung 4 – Spaltenverschub mit Definition der Höhe



Ausgangsbasis ist Zelle D3 aus *Bezug*.


Ein Zeilenverschub erfolgt nicht (keine Angabe in *Zeilen* bedeutet Null) – wir bleiben in Zelle D3. Das Funktionsargument *Spalten* enthält eine 2 – wir verschieben damit den Bezug um 2 Spalten nach rechts, also auf Zelle F3.

Das Argumente *Höhe* ist mit 10 angegeben: der Zielbereich umfasst also 10 Zeilen, beginnend bei Zelle F3 – das ist der Bereich F3:F12.

Das Argument *Breite* ist nicht angegeben, es bleibt also bei der Dimensionierung aus Bezug (das war die eine Spalte D).

Der Zielbereich umfasst die Zellen: **F3:F12**.

Übung 5 – Zeilen-/Spaltenverschub mit negativer Dimension



Ausgangsbasis ist Zelle B3 aus *Bezug*.

Nun wird um 1 Zeile (nach B4) und um 2 Spalten (nach D4) verschoben.

Das Argumente *Höhe* ist hier mit **minus** 3 angegeben: Der Zielbereich wird hier nun nicht nach unten, sondern nach oben (!) auf insgesamt 3 Zeilen erweitert – ab Zelle D4, insgesamt 3 Zellen nach oben, ergibt den Bereich D2:D4.

Das Argument *Breite* ist mit **minus** 2 definiert – der Zielbereich wird um insgesamt 2 Spalten nach links (!) erweitert. – Der Zielbereich umfasst die Zellen: **C2:D4**

	A	B	C	D	E	F	G	H
1	11	21	31	41	51			
2	12	22	32	42	52			
3	13	23	33	43	53			
4	14	24	34	44	54			
5	15	25	35	45	55			

228 =SUMME(BEREICH.VERSCHIEBEN(B3;1;2;-3;-2))

Erläuterungen

einzeiliger Zielbereich (Beispiel 1)

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3			1.000	=BEREICH.VERSCHIEBEN(A1;4;2;1;1)
3	11	Willi	10	20	30				
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000				
6	14	Ralf	10.000	20.000	30.000				
7	15	Paul	100.000	200.000	300.000				

=BEREICH.VERSCHIEBEN(A1;4;2;1;1)

... in Zelle H2 definiert den Bereich (bzw. hier die einzelne Zelle) **C5**.

In einfachen Worten: gehe von Zelle A1 aus 4 Zeilen nach unten und 2 Spalten nach rechts. Höhe und Breite sind jeweils mit 1 definiert und beschreiben so einen einzeiligen und einspaltigen Bezug (eine Zelle).

Weil bereits die Angabe von „A1“ in *Bezug* einen einspaltigen und einzeiligen Bereich definiert, hätte die Angabe für die Funktionsargumente *Höhe* und *Breite* in der Funktion oben entfallen können – es hätte so genügt:

=BEREICH.VERSCHIEBEN(A1;4;2)

mehrzelliger Zielbereich (Beispiel 2)

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3			1.000	=BEREICH.VERSCHIEBEN(A1;4;2;1;1)
3	11	Willi	10	20	30			#WERT!	=BEREICH.VERSCHIEBEN(A1;4;2;1;3)
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000				
6	14	Ralf	10.000	20.000	30.000				
7	15	Paul	100.000	200.000	300.000				

=BEREICH.VERSCHIEBEN(A1;4;2;1;3)

... in Zelle H3 definiert einen 3-spaltigen Bereich, und zwar **C5:E5**.

Ergebnis der Formel in Zelle H3 ist der Fehler #WERT!.

Ein Bereich ist ein Bereich

Beispiel 1 zeigt, dass BEREICH.VERSCHIEBEN in der Lage ist, den Inhalt einer Zelle wiederzugeben, sofern der durch die Funktion ermittelte Zellbezug aus nur **einer** Zelle besteht.

Besteht der Bereich aus mehreren Zellen (siehe Beispiel 2: Zellen C5:E5), benötigt die Funktion BEREICH.VERSCHIEBEN eine weitere Funktion, mit der bestimmt wird, was mit diesem Bereich *anzufangen* ist.

BEREICH.VERSCHIEBEN ist in fast jeder integrierten Excelfunktion als "Bezug" nutzbar. Siehe dazu folgende Beispiele 3 und 4.

SUMME mit BEREICH.VERSCHIEBEN (Beispiel 3)

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3			Summe	
3	11	Willi	10	20	30			6.000	=SUMME(BEREICH.VERSCHIEBEN(A1;4;2;1;3))
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000			1.000	=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);1)
6	14	Ralf	10.000	20.000	30.000			2.000	=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);2)
7	15	Paul	100.000	200.000	300.000			3.000	=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);3)

=SUMME(BEREICH.VERSCHIEBEN(A1;4;2;1;3))

... in Zelle H3 ermittelt die Summe des gewählten Bereiches C5:E5.

INDEX mit BEREICH.VERSCHIEBEN (Beispiel 4)

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3			Summe	
3	11	Willi	10	20	30			6.000	=SUMME(BEREICH.VERSCHIEBEN(A1;4;2;1;3))
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000			1.000	=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);1)
6	14	Ralf	10.000	20.000	30.000			2.000	=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);2)
7	15	Paul	100.000	200.000	300.000			3.000	=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);3)

=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);1)

=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);2)

=INDEX(BEREICH.VERSCHIEBEN(A1;4;2;1;3);3)

... in Zellen H5 bis H7 gibt die einzelnen Werte des Bereiches aus.

Dynamisch verschieben (Beispiel 5)

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3		Nr.	Summe	
3	11	Willi	10	20	30		11	60	=SUMME(BEREICH.VERSCHIEBEN(A1;VERGLEICH(G3;A:A;0)-1;2;1;3))
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000				
6	14	Ralf	10.000	20.000	30.000				
7	15	Paul	100.000	200.000	300.000				

=SUMME(BEREICH.VERSCHIEBEN(A1;VERGLEICH(G3;A:A;0)-1;2;1;3))

... in Zelle H3 ermittelt die Summe des Bereiches C3:E3. Allerdings ist der Bereich hier variabel gehalten: Die Zeile des mit BEREICH.VERSCHIEBEN definierten Bereiches wird nun mit der Funktion VERGLEICH ermittelt, und VERGLEICH verwendet das Suchkriterium in Zelle G3.

Es genügt hier den Eintrag in Zelle G3 beispielsweise in „13“ zu verändern, um die Summe für die Zellen C5:E5 zu erhalten:

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3		Nr.	Summe	
3	11	Willi	10	20	30		13	6.000	=SUMME(BEREICH.VERSCHIEBEN(A1;VERGLEICH(G3;A:A;0)-1;2;1;3))
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000				
6	14	Ralf	10.000	20.000	30.000				
7	15	Paul	100.000	200.000	300.000				

VERGLEICH liefert die Positionsnummer des Suchkriteriums (hier „13“ aus Zelle G3) innerhalb des angegebenen Bereiches (hier die komplette Spalte A, da "A:A"). Ergebnis von VERGLEICH ist hier

5. Ausgehend von Zelle A1 wollen wir den Bereich nicht um 5 Zeilen (nach A6) verschieben, sondern um 4 Zeilen (nach A5). Dafür wurde 1 vom Ergebnis der Funktion VERGLEICH subtrahiert. Das ist erforderlich, weil Ausgangsbasis unseres Bereich-Versatzes die Zelle A1 ist – wir befinden uns bereits in Zelle 1. Eine Zelle A0 kann nicht Ausgangsbasis sein, weil es diese Zelle A0 nicht gibt.

Performance

In Excel gibt es stets mehrere Lösungswege, die sich oft in ihrer Performance unterscheiden. Bedeutend dafür ist meist die Größe des zu berechnenden Bereiches bzw. die Anzahl der Berechnungen überhaupt.

Im vorangegangenen Beispiel haben wir mit der Formel

=SUMME(BEREICH.VERSCHIEBEN(A1;VERGLEICH(G3;A:A;0)-1;2;1;3))

die Summe der Zellen C5:E5 gebildet (G3 enthält den Eintrag „13“). Dabei haben wir innerhalb der Funktion VERGLEICH die gesamte Spalte A für den Bezug zur Verfügung, d. h. die Funktion ist für alle 65536 Zeilen anwendbar. Und sie summiert dennoch nur diese drei Zellen!

Sollte diese Berechnung mit der Funktion SUMMENPRODUKT durchgeführt werden, müsste es so aussehen:

=SUMMENPRODUKT((A2:A65536=G3)*1;C2:C65536+D2:D65536+E2:E65536)

Hier sind stets, egal wie viele Zeilen Sie tatsächlich belegt haben, 65535 Zeilen zu berechnen, und zwar die Prüfung der Zellen aus Spalte A und die Multiplikation der Wahrheitswerte aus dieser Prüfung mit den jeweiligen Zelleninhalten der Zellen der Spalten C und D und E. Das wären 4 mal 65535 Berechnungen.

Während BEREICH.VERSCHIEBEN hier lediglich 3 Zellen betrachtet, *schaut* sich SUMMENPRODUKT gut 260.000 Zellen an! Und eben dies wirkt sich (z. T. ganz erheblich) auf die Performance aus.

Bereich „verkleinern“ (Beispiel 6)

Wenn Sie die Daten in den Zellen A1:J12 nachbilden wollen – Inhalt der Zellen ist die Formel

=SPALTE()*10+ZEILE()

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	11	21	31	41	51	61	71	81	91	101			
2	12	22	32	42	52	62	72	82	92	102			
3	13	23	33	43	53	63	73	83	93	103			
4	14	24	34	44	54	64	74	84	94	104			
5	15	25	35	45	55	65	75	85	95	105			
6	16	26	36	46	56	66	76	86	96	106			
7	17	27	37	47	57	67	77	87	97	107			
8	18	28	38	48	58	68	78	88	98	108			
9	19	29	39	49	59	69	79	89	99	109			
10	110	210	310	410	510	610	710	810	910	1010			
11	111	211	311	411	511	611	711	811	911	1011			
12	112	212	312	412	512	612	712	812	912	1012			

Die Formel (hier in zwei alternativen Schreibweisen):

=SUMME(BEREICH.VERSCHIEBEN(B3:E5;;;1;3))

=SUMME(BEREICH.VERSCHIEBEN(B3:E5;0;0;1;3))

... verschiebt den Bereich B3:E5 nicht (keine Angabe von Zeilen- oder Spaltenverschiebung bzw. Null), sondern reduziert ihn (1.) auf eine Zeile (da wären wir bei B3:E3) und (2.) auf drei Spalten (so gelangen wir zu B3:D3).

Die Formel ist oben in Zelle L2 angewendet und zeigt die Summe der Zellen B3:D3.

Es ist nicht ganz korrekt, von „Verkleinern“ des Bereiches zu sprechen, vielmehr wird der Bereich – ausgehend von der ersten Zelle des verschobenen Bereiches – **anders dimensioniert**.

gleich dimensionierter Bereich (Beispiel 7)

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	11	21	31	41	51	61	71	81	91	101			
2	12	22	32	42	52	62	72	82	92	102		3228	=SUMME(BEREICH.VERSCHIEBEN(B3:E5;5;3))
3	13	23	33	43	53	63	73	83	93	103			
4	14	24	34	44	54	64	74	84	94	104			
5	15	25	35	45	55	65	75	85	95	105			
6	16	26	36	46	56	66	76	86	96	106			
7	17	27	37	47	57	67	77	87	97	107			
8	18	28	38	48	58	68	78	88	98	108			
9	19	29	39	49	59	69	79	89	99	109			
10	110	210	310	410	510	610	710	810	910	1010			
11	111	211	311	411	511	611	711	811	911	1011			
12	112	212	312	412	512	612	712	812	912	1012			

=SUMME(BEREICH.VERSCHIEBEN(B3:E5;5;3))

... verschiebt den Bereich B3:E5 um 5 Zeilen (da wären wir bei B8:E10) und um 3 Spalten (so gelangen wir zu dem Bereich E8:H10). Die Formel ist oben in Zelle L2 angewendet und zeigt die Summe der Zellen E8:H10. Oben grün markiert.

Dynamische Bereiche

Dynamische Gültigkeitsliste (Beispiel 8)

Mit der Funktion BEREICH.VERSCHIEBEN können Sie einen dynamischen Bereich für die Funktionalität → Daten → Gültigkeit definieren.

	A	B	C	D
1	Name			
2	Hans		Hans	
3	Willi		Hans	
4	Fritz		Willi	
5			Fritz	
6				


Ergänzen wir die Liste in Spalte A um weitere Vornamen, erhalten wir in Zelle C2 eine neue Auswahl – stets so viele, wie es zur Auswahl stehende Einträge in Spalte A gibt.

	A	B	C	D
1	Name			
2	Hans		Hans	
3	Willi		Hans	
4	Fritz		Willi	
5	Christian		Fritz	
6	Ralf		Christian	
7	Paul		Ralf	
8			Paul	

Wie wird's gemacht?

Zunächst haben wir einen Namen definiert:

- Einfügen
- Namen
- Definieren...




... mit „Auswahl“ benannt und die Formel unter *Bezieht sich auf:* eingetragen. (In unserem Fall lautet der Tabellenname „Tab5“).

Tip 1: geben Sie die Formel OHNE Tabellenblattnamen ein und klicken Sie dann auf *Hinzufügen*. Automatisch wird allen Zellbezügen der aktuelle Tabellenblattname vorangestellt.

Tip 2: Bei der Eingabe im Feld *Bezieht sich auf:* befindet sich Excel im "Eingeben-Modus". Dieser sorgt dafür, dass beim Drücken einer der Cursor- bzw. Pfeil-Tasten auf der Tastatur sich die Zellmarkierung verändert und die Adresse der aktiven Zelle in die Formel eingetragen wird. Dieses Verhalten wird in den seltensten Fällen erwünscht, eher ist es störend. Abhilfe schafft das einmalige Drücken der Taste F2, denn erst dadurch wechselt Excel in den "Bearbeiten-Modus". Nun kann man wie gewohnt innerhalb der Formel navigieren.

Für Zelle C2 haben wir danach die Gültigkeit angewendet:

- Daten
- Gültigkeit



Voilà!

Relative Bezüge

	A	B	C	D	E	F	G	H	I	J
1	11	21	31	41	51	61	71			
2	12	22	32	42	52	62	72			
3	13	23	33	43	53	63	73			
4	14	24	34	44	54	64	74		333	=SUMME(BEREICH.VERSCHIEBEN(B2:D3;3;2))
5	15	25	35	45	55	65	75			
6	16	26	36	46	56	66	76			
7	17	27	37	47	57	67	77			
8	18	28	38	48	58	68	78			

Das kennen Sie bereits: Die Formel in Zelle I4

=SUMME(BEREICH.VERSCHIEBEN(B2:D3;3;2))

liefert die Summe für die Zellen D5:F6.

Wir verwenden hier einen relativen Bezug auf die Zellen B2:D3.

Folge daraus:

- Kopieren wir die Formel von I4 (Strg+C) nach I6 (Strg+V), verändert sich Bezug auf B4:D5, folglich liefert uns die Formel nun die Summe der Zellen D7:F8 (in folgender Grafik in der Farbe Magenta dargestellt).
- Kopieren wir die Formel von I4 nach I2, wird uns der Fehler #BEZUG! gemeldet. Die Änderung um zwei Zeilen *nach oben* wählte einen Bezug kleiner Zeile 1, und diesen Bereich gibt es nicht.


	A	B	C	D	E	F	G	H	I	J
1	11	21	31	41	51	61	71			
2	12	22	32	42	52	62	72		#BEZUG!	=SUMME(BEREICH.VERSCHIEBEN(#BEZUG!;3;2))
3	13	23	33	43	53	63	73			
4	14	24	34	44	54	64	74		333	=SUMME(BEREICH.VERSCHIEBEN(B2:D3;3;2))
5	15	25	35	45	55	65	75			
6	16	26	36	46	56	66	76		345	=SUMME(BEREICH.VERSCHIEBEN(B4:D5;3;2))
7	17	27	37	47	57	67	77			
8	18	28	38	48	58	68	78			

Vermutlich war Ihnen das schon bewusst, zumindest aber ist es für Sie gewiss nicht überraschend. Und warum schreiben wir es hier dennoch? Um Ihnen zu berichten, dass eben dieses Prinzip selbstredend auch dann gilt, wenn Sie einen relativen Bezug innerhalb eines Namens definieren. Und die versehentliche Nichtbeachtung dieser *Selbstverständlichkeit* veranlasst zuweilen dazu, die *höchsten Palmen zu besteigen*. Und eben davor wollen wir Sie bewahren – jawohl, so nett sind wir!

Bei Definition des Namens „Auswahl2“ befanden wir uns in der (aktiven) Zelle I4, wählten

- Einfügen
- Namen
- Definieren

und trugen die Daten ein (siehe Grafik):



In Zelle I4 bildeten wir nun die Summe mit

=SUMME(Auswahl2)


	A	B	C	D	E	F	G	H	I	J
1	11	21	31	41	51	61	71			
2	12	22	32	42	52	62	72			
3	13	23	33	43	53	63	73			
4	14	24	34	44	54	64	74		333	=SUMME(Auswahl2)
5	15	25	35	45	55	65	75			
6	16	26	36	46	56	66	76			
7	17	27	37	47	57	67	77			
8	18	28	38	48	58	68	78			

Wenn wir nun die Formel aus I4 nach I6 und I2 kopieren, erhalten wir die gleichen Ergebnisse wie oben:

	A	B	C	D	E	F	G	H	I	J
1	11	21	31	41	51	61	71			
2	12	22	32	42	52	62	72		#BEZUG!	=SUMME(Auswahl2)
3	13	23	33	43	53	63	73			
4	14	24	34	44	54	64	74		333	=SUMME(Auswahl2)
5	15	25	35	45	55	65	75			
6	16	26	36	46	56	66	76		345	=SUMME(Auswahl2)
7	17	27	37	47	57	67	77			
8	18	28	38	48	58	68	78			

Selbstverständlich kann eine vorstehend aufgezeigte Bereichswahl mit den relativen Bezügen gewollt sein. Sehr häufig aber wohl nicht. Und insbesondere in Verbindung mit NAMEN dürfte der Wunsch hierzu gegen Null streben, denn hier ist die Bereichswahl auch noch von der aktiven Zelle abhängig, in der wir uns befanden, als wir den NAMEN definierten. – Die Alternative besteht in absoluten Bezügen.

Absolute Bezüge



=BEREICH.VERSCHIEBEN(Tabelle10!\$B\$2:\$D\$3;3;2)

	A	B	C	D	E	F	G	H	I	J
1	11	21	31	41	51	61	71			
2	12	22	32	42	52	62	72		333	=SUMME(Auswahl3)
3	13	23	33	43	53	63	73			
4	14	24	34	44	54	64	74		333	=SUMME(Auswahl3)
5	15	25	35	45	55	65	75			
6	16	26	36	46	56	66	76		333	=SUMME(Auswahl3)
7	17	27	37	47	57	67	77			
8	18	28	38	48	58	68	78			

Ein typischer Anwendungsfall ergibt sich in Verbindung mit INDEX:

	A	B	C	D	E	F	G	H	I
1	Nr.	Name	2007	2008	2009				
2	10	Hans	1	2	3				
3	11	Willi	10	20	30				
4	12	Fritz	100	200	300				
5	13	Christian	1.000	2.000	3.000		1.000	=INDEX(BEREICH.VERSCHIEBEN(\$A\$1;4;2;1;3);ZEILE(A1))	
6	14	Ralf	10.000	20.000	30.000		2.000	=INDEX(BEREICH.VERSCHIEBEN(\$A\$1;4;2;1;3);ZEILE(A2))	
7	15	Paul	100.000	200.000	300.000		3.000	=INDEX(BEREICH.VERSCHIEBEN(\$A\$1;4;2;1;3);ZEILE(A3))	

Die Funktion

=INDEX(BEREICH.VERSCHIEBEN(\$A\$1;4;2;1;3);ZEILE(A1))

haben wir in Zelle H5 geschrieben und nach unten kopiert.